RF power transformer design

This is a pretty complex business and I propose to deal with the 'how' rather than 'why'. You need, at minimum, to know the following things about the ferrite material. I have assumed the use of ferrite rings typical of the power designs usually published. I have also assumed that one winding of the transformer will operate to or from a 50 ohm load. The other winding is dictated purely by the transformation requirements. **You must know:**

The outer diameter of the ring: Do (mm)
The inner diameter of the ring: d(mm)
The height of the ring: h (mm)
The relative permeability: μr
The turns ratio of the transformers: n (this assumes that the secondary, the collector or base winding, will comprise a single turn made out of copper or brass tube, and that n will be the number of turns on the primary for connection to a 50 ohm load)
The minimum operating frequency: Fmin (MHz)
The power to be transformed: Po (Watts)
The power loss per cubic cm of ferrite material at a given frequency and specified magnetic induction: Ploss (mW)

The last two quantities need only be known if loss calculations are to be made.

Other quantities: μo (permeability of free space) $= 4 \times Pi \times 10^{-7}$

First calculate the cross sectional area of an individual core: s (m^2)

$s = h \times (Do - d)/2 \times 10^{-6}$ square metres



Fig. 7. If a transistor PA is to develop its full power over the range 1.8 to 30MHz, the transformer leakage inductance must be carefully cancelled out over the same frequency span. The inductance is made to form part of a 'T' filter arrangement with a characteristic cut-off frequency just above 30MHz.

Next, calculate the magnetic path length: l

$l = (Do - ((Do - d)/2)) \times Pi$ mm

Now you need to know the minimum inductance which the circuit design can tolerate. Generally, a minimum reactance of 150 ohms will be adequate for a 50 ohm system.

$Lp = 150/2 \times Pi \times Fmin\ \mu H$

The number of ferrite cores, Nc, required to achieve the desired performance is:

$Nc = (Lp \times l \times 10^{-9}) / (\mu o \times \mu r \times n^2 \times s)$

Round the value Nc up to the nearest whole even number.

To work out the system losses

v, the maximum peak voltage across the primary (50 ohms) is calculated:

v = square root of (Po x 100)

Work out the maximum induction within the core: B (mT)

$B = 10 \times (v/(2 \times Pi \times Fmin \times 10^3 \times n \times Nc \times s))$ **mT . . . milliTesla**

Next work out the volume of the core in cm^3

$Vol = Pi/4 \times (Do^2 - d^2) \times h \times Nc \times 10^{-3}$

The total power lost in the core = vol x Ploss mW

I have devised a computer program suitable for the Sinclair Spectrum which takes the chore out of transformer design. The listings and program output have been included. For guidance, the typical loss figure for ferrite materials with a μr in the region of 100 to 200 is 200mW/cm^3 at a maximum induction of 12mT at 1.8MHz.

```
   5 REM "Broadband"
  10 PRINT "Enter details of tor
oid cores"
  15 PRINT "All dimensions to be
 in mm"
  20 INPUT "Specify outer diamet
er ";Do
  30 CLS
  40 INPUT "Inner diameter ";d
  50 CLS
  60 INPUT "And the height ";h
  70 INPUT "Specify  relative pe
rmeability ",ur
  80 PRINT "Next, the turns rati
o"
  90 INPUT "How many turns on pr
imary? ";n
 100 CLS
 110 LET uo=4*PI*10↑-7
 120 LET s=h*((Do-d)/2)*10↑-6
 130 LET l=(Do-((Do-d)/2))*PI
 140 PRINT "Primary inductance"
 145 PRINT "This assumes reactan
ce","is equal to three times ","
operating impedance normalised",
"to 50 ohms"
 150 INPUT "Specify lowest frequ
ency MHz ",Fmin
 155 CLS
 160 LET Lp=INT (150/(2*PI*Fmin)
): PRINT "Primary inductance = "
;Lp;" uH"
 170 LET Nc=INT ((Lp*l*10↑-9)/(u
o*ur*n↑2*s)+.5)
 175 PRINT
 180 PRINT "[illegible]
[illegible]:"
 185 PRINT
 186 PRINT "Outer diameter ";Do;
"mm"
 187 PRINT
 188 PRINT "Inside diameter ";d;
"mm"
 189 PRINT
 190 PRINT "Height ";h;"mm"
 191 PRINT
 192 PRINT "Relative permeabilit
y ";ur
 193 PRINT
 194 PRINT "Minimum frequency ";
Fmin;"MHz"
 195 PRINT
 200 PRINT "Number of primary tu
rns = ";n
 205 PRINT
 220 PRINT "Number of cores = ";
Nc
 225 PRINT
 230 INPUT "Enter transferred po
wer (W)","assuming a 50 ohm matc
hed ","load ";Po
 240 LET v=SQR (2*Po*50)
 250 LET B=INT (10*(v/(2*PI*Fmin
*10↑3*n*Nc*s)))/10: PRINT "Max i
nduction = ";B;"mT"
 260 INPUT "Power loss density o
f ferrite","material at the maxi
mum flux","induction level and t
he minimum","frequency: mW/cm↑3
";Ploss
 270 LET vol=PI/4*(Do↑2-d↑2)*h*N
c*10↑-3
 280 PRINT
 290 PRINT "Power loss in core i
s ";INT (Ploss*vol);"mW","at ";P
o;"W transferred power level"
```

Fig. 8. Sinclair Spectrum transformer design program.

```
[illegible]:

Outer diameter 12.7mm

Inside diameter 6.35mm

Height 6.35mm

Relative permeability 220

Minimum frequency 1.8MHz

Number of primary turns = 3

Number of cores = 8

Max induction = 12.9mT

Power loss in core is 865mW
at 50W transferred power level
```

Fig. 9. Program output (above)



Fig. 10. The two computer plots (above) show the effect of output resistance changes on frequency response for the circuit shown in Fig. 3.